################################################################################
# Copyright (C) 2014-2022 GSI Helmholtzzentrum fuer Schwerionenforschung GmbH  #
#                                                                              #
#              This software is distributed under the terms of the             #
#              GNU Lesser General Public Licence (LGPL) version 3,             #
#                  copied verbatim in the file "LICENSE"                       #
################################################################################

set(target ExPixelAlt)

set(no_dict_sources
  devices/FairMQPixAltFileSinkBin.cxx
  devices/FairMQPixAltSamplerBin.cxx
)

fair_change_extensions_if_exists(.cxx .h FILES "${no_dict_sources}" OUTVAR no_dict_headers)

list(APPEND no_dict_headers
  devices/FairMQPixAltTaskProcessorBin.h
)

set(sources
  PixelAltDigiWriteToRootVector.cxx
  PixelAltFindHits.cxx
)

fair_change_extensions_if_exists(.cxx .h FILES "${sources}" OUTVAR headers)

add_library(${target} SHARED ${sources} ${no_dict_sources} ${headers} ${no_dict_headers})
add_library(FairRoot::${target} ALIAS ${target})
set_target_properties(${target} PROPERTIES ${PROJECT_LIBRARY_PROPERTIES})

target_include_directories(${target} PUBLIC
  $<BUILD_INTERFACE:${CMAKE_CURRENT_SOURCE_DIR}>
  $<BUILD_INTERFACE:${CMAKE_CURRENT_SOURCE_DIR}/devices>
  $<INSTALL_INTERFACE:${CMAKE_INSTALL_INCLUDEDIR}>
)

target_link_libraries(${target}
  PUBLIC
  FairRoot::Base # FairRunIdGenerator
  FairRoot::ParBase
  FairRoot::GeoBase
  FairRoot::Tools # FairLogger
  FairRoot::BaseMQ # Serialization policies
  FairRoot::FairMQ
  FairRoot::ExPixel

  ROOT::Core

  PRIVATE
  FairLogger::FairLogger

  ROOT::Tree
  ROOT::MathCore # Sqrt
  ROOT::Geom # TGeo...
)

fairroot_target_root_dictionary(${target}
  HEADERS ${headers}
  LINKDEF PixelAltLinkDef.h
  EXTRA_INCLUDE_DIRS ${CMAKE_INSTALL_FULL_INCLUDEDIR}
)

install(TARGETS ${target} LIBRARY DESTINATION ${PROJECT_INSTALL_LIBDIR})
install(FILES ${headers} ${no_dict_headers} DESTINATION ${CMAKE_INSTALL_INCLUDEDIR})
